
Advisor Answers
Visual FoxPro 5.0 and 3.0 and FoxPro 2.x
Q: Does the SELECT statement support creation of empty memo fields as it supports
creating empty character fields?
For example, consider:
SELECT SPACE(10) AS TAG, ;
???? AS MY_MEMO, ;
NAME, ADDR ;
FROM SRC_DBF ;
INTO TABLE TARG_TBL ;
WHERE...

What is the syntax of "???? AS MY_MEMO" where MY_MEMO is to be a memo field in the
target table?
The workaround I have been using is to get the memo field from another table that has
one blank memo field record and join the two tables in the WHERE clause. But a more
direct method would be much faster, I'm sure.
–Arden Weiss (via Advisor.Com)
A: Actually, there is no direct way to create empty memo fields using SELECT. There's
also no direct way to create empty fields of type Integer or Double.
The work-around you're using is very close to what I recommend. Rather than keeping a
table around for this purpose, I suggest you use a cursor that you create on the fly:
CREATE CURSOR Dummy (My_Memo M)

Add one empty record to it:
APPEND BLANK

Then you can use this cursor in the query. You don't put any special code in the WHERE
clause because you're doing what's known as a Cartesian join. Normally, a Cartesian
join, where every record of one table is joined with every record of another, is a terrible
idea. But, in this case, where the cursor contains only a single record, it gives you
exactly the results you want.
SELECT SPACE(10) AS TAG, ;
My_Memo AS MY_MEMO, ;
NAME, ADDR ;
FROM SRC_DBF, Dummy ;
INTO TABLE TARG_TBL ;
WHERE...

Finally, when you're done, clean up:
USE IN Dummy

As I noted above, integer and double present the same problem and lend themselves to
the same solution. For other field types, use SPACE(), as shown in the example, for
character fields, {} for empty date fields, TTOC(" ") to create datetimes, .F. to create
logicals and $0 to create currency.
–Tamar

